pragma solidity ^0.4.14;

/**
 * @title SafeMath
 * @dev Math operations with safety checks that throw on error
 */
library SafeMath {
  function mul(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a * b;
    assert(a == 0 || c / a == b);
    return c;
  }

  function div(uint256 a, uint256 b) internal constant returns (uint256) {
    // assert(b > 0); // Solidity automatically throws when dividing by 0
    uint256 c = a / b;
    // assert(a == b * c + a % b); // There is no case in which this doesn't hold
    return c;
  }

  function sub(uint256 a, uint256 b) internal constant returns (uint256) {
    assert(b <= a);
    return a - b;
  }

  function add(uint256 a, uint256 b) internal constant returns (uint256) {
    uint256 c = a + b;
    assert(c >= a);
    return c;
  }
}

/**
 * @title ERC20Basic
 * @dev Simpler version of ERC20 interface
 * @dev see https://github.com/ethereum/EIPs/issues/179
 */
contract ERC20Basic {
  uint256 public totalSupply;
  function balanceOf(address who) public constant returns (uint256);
  function transfer(address to, uint256 value) public returns (bool);
  event Transfer(address indexed from, address indexed to, uint256 value);
}

/**
 * @title Basic token
 * @dev Basic version of StandardToken, with no allowances.
 */
contract BasicToken is ERC20Basic {
  using SafeMath for uint256;

  mapping(address => uint256) balances;
  //Tracks Xmas tokens that have been used to claim from the giftPool
  mapping(address => uint256) usedTokens;

  /**
  * @dev transfer token for a specified address
  * @param _to The address to transfer to.
  * @param _value The amount to be transferred.
  */
  function transfer(address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[msg.sender]);

    // SafeMath.sub will throw if there is not enough balance.
    balances[msg.sender] = balances[msg.sender].sub(_value);
    balances[_to] = balances[_to].add(_value);
    if(1 <= usedTokens[msg.sender]) { 
        usedTokens[msg.sender] = usedTokens[msg.sender].sub(_value);
        usedTokens[_to] = usedTokens[_to].add(_value);
    }
    Transfer(msg.sender, _to, _value);
    return true;
  }

  /**
  * @dev Gets the balance of the specified address.
  * @param _owner The address to query the the balance of.
  * @return An uint256 representing the amount owned by the passed address.
  */
  function balanceOf(address _owner) public constant returns (uint256 balance) {
    return balances[_owner];
  }

}

contract ERC20 is ERC20Basic {
  function allowance(address owner, address spender) public constant returns (uint256);
  function transferFrom(address from, address to, uint256 value) public returns (bool);
  function approve(address spender, uint256 value) public returns (bool);
  event Approval(address indexed owner, address indexed spender, uint256 value);
}

/**
 * @title Standard ERC20 token
 *
 * @dev Implementation of the basic standard token.
 * @dev https://github.com/ethereum/EIPs/issues/20
 * @dev Based on code by FirstBlood: https://github.com/Firstbloodio/token/blob/master/smart_contract/FirstBloodToken.sol
 */
contract StandardToken is ERC20, BasicToken {

  mapping (address => mapping (address => uint256)) internal allowed;


  /**
   * @dev Transfer tokens from one address to another
   * @param _from address The address which you want to send tokens from
   * @param _to address The address which you want to transfer to
   * @param _value uint256 the amount of tokens to be transferred
   */
  function transferFrom(address _from, address _to, uint256 _value) public returns (bool) {
    require(_to != address(0));
    require(_value <= balances[_from]);
    require(_value <= allowed[_from][msg.sender]);

    balances[_from] = balances[_from].sub(_value);
    usedTokens[_from] = usedTokens[_from].sub(_value);
    balances[_to] = balances[_to].add(_value);
    usedTokens[_to] = usedTokens[_to].add(_value);
    allowed[_from][msg.sender] = allowed[_from][msg.sender].sub(_value);
    Transfer(_from, _to, _value);
    return true;
  }

  /**
   * @dev Approve the passed address to spend the specified amount of tokens on behalf of msg.sender.
   *
   * Beware that changing an allowance with this method brings the risk that someone may use both the old
   * and the new allowance by unfortunate transaction ordering. One possible solution to mitigate this
   * race condition is to first reduce the spender's allowance to 0 and set the desired value afterwards:
   * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
   * @param _spender The address which will spend the funds.
   * @param _value The amount of tokens to be spent.
   */
  function approve(address _spender, uint256 _value) public returns (bool) {
    allowed[msg.sender][_spender] = _value;
    Approval(msg.sender, _spender, _value);
    return true;
  }

  /**
   * @dev Function to check the amount of tokens that an owner allowed to a spender.
   * @param _owner address The address which owns the funds.
   * @param _spender address The address which will spend the funds.
   * @return A uint256 specifying the amount of tokens still available for the spender.
   */
  function allowance(address _owner, address _spender) public constant returns (uint256 remaining) {
    return allowed[_owner][_spender];
  }

  /**
   * approve should be called when allowed[_spender] == 0. To increment
   * allowed value is better to use this function to avoid 2 calls (and wait until
   * the first transaction is mined)
   * From MonolithDAO Token.sol
   */
  function increaseApproval (address _spender, uint _addedValue) public returns (bool success) {
    allowed[msg.sender][_spender] = allowed[msg.sender][_spender].add(_addedValue);
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

  function decreaseApproval (address _spender, uint _subtractedValue) public returns (bool success) {
    uint oldValue = allowed[msg.sender][_spender];
    if (_subtractedValue > oldValue) {
      allowed[msg.sender][_spender] = 0;
    } else {
      allowed[msg.sender][_spender] = oldValue.sub(_subtractedValue);
    }
    Approval(msg.sender, _spender, allowed[msg.sender][_spender]);
    return true;
  }

}


contract XmasCoin is StandardToken {
  
  using SafeMath for uint;

  string public constant name = "XmasCoin";
  string public constant symbol = "XMAS";
  uint public constant decimals = 0;
  uint public constant totalSupply = 100;

  address public owner = msg.sender;
  //Ether that will be gifted to all token holders after the deadline passes 
  uint public giftPool;
  //Ether that will be used for the next holiday gift pool after the deadline passes 
  uint public timeToOpenPresents;

  event GiftPoolContribution(address giver, uint amountContributed);
  event GiftClaimed(address claimant, uint amountClaimed, uint tokenAmountUsed);

  modifier onlyOwner {
      require(msg.sender == owner);
      _;
  }

  function XmasCoin() {
    //issue coins to owner for holiday distribution
    balances[owner] = balances[owner].add(totalSupply);

    //gift time set to Christmas Day, December 25, 2017 12:00:00 AM GMT
    timeToOpenPresents = 1514160000;
  } 

  function claimXmasGift(address claimant)
    public
    returns (bool)
  {
    require(now > timeToOpenPresents);
    require(1 <= validTokenBalance(claimant));
    
    uint amount = giftBalance(claimant);
    uint tokenBalance = validTokenBalance(claimant);
    usedTokens[claimant] += tokenBalance;
    
    claimant.transfer(amount);
    GiftClaimed(claimant, amount, tokenBalance);
    
    return true;
  }

  //all eth sent to fallback is added to the gift pool, stops accepting when presents open
  function () 
    public
    payable
  {
    require(msg.value > 0);
    require(now < timeToOpenPresents);
    giftPool += msg.value;
    GiftPoolContribution(msg.sender, msg.value);
  } 
  
  function validTokenBalance (address _owner)
    public
    constant
    returns (uint256)
  {
      return balances[_owner].sub(usedTokens[_owner]);
  }
  
  function usedTokenBalance (address _owner)
    public
    constant
    returns (uint256)
  {
      return usedTokens[_owner];
  }
  
  function giftBalance(address claimant)
    public
    constant
    returns (uint)
  {
    return giftPool.div(totalSupply).mul(validTokenBalance(claimant));    
  }
  
  function selfDestruct()
    public
    onlyOwner
  {
     suicide(owner);
  }
  
}